Logitek Application Note
Profanity Delay Status

Applies to Audio Engine with AE-C6 v3.61 & up with SHARC Attack DSP and all JetStream routers.
Last updated 14 June 2011.

Starting with September 2005's AE-C6 v3.61, we have three busses that help keep track of the SHARC Attack profanity delay's status. (While the AE-C2 card, paired with a SHARC Attack will do delay, these extra signaling busses are not available with the AE-C2 card.) This feature was ported in its entirety to the JetStream and functions in the same manner.

By hanging a trigger on these busses, you can add text to your console display or have the lights flash to show what state the delay is in.

First, a brief overview of the busses:

Device number = the router crosspoint that is being used for the delay. Usually this is Port 1 Router 1 (device 30).

Bus 24 - turns on when the delay is activated and ramping up. This bus stays on while the delay is active.

Bus 25 - turns on when the delay is ramping down. (Bus 24 and 26 will turn off when this turns on.) Bus 25 will turn off when the delay reaches zero.

Bus 26 - turns on when the delay has reached maximum duration and turns off when the DUMP command is issued.

Command Builder Example Triggers for Profanity Delay

For the purposes of this article, the delay ramps in and out with a toggle switch on bridge button 1 and the dump button is on bridge button 2. Port 1 Router 1 will be used for the delay. Your implementation of the delay may vary. Any text after a tilde (~) is a comment and is ignored by Command Builder.

Trigger #1: Delay Start and Stop

```
trigger ae1 surface 1 bridge button 1 on toggle

      if toggle = 1
        cmd ae1 d[Port1 Router 1 In] talk max 10.0 ~ sets delay buffer to 10.0 secs
        cmd ae1 d[Port1 Router 1 In] talk start ~ starts delay
      endif

      if toggle = 2
        cmd ae1 d[Port1 Router 1 In] talk stop ~ ramps to zero
      endif
```

Trigger #2 Delay Dump

```
trigger ae1 surface 1 bridge button 2 on

      cmd ae1 d[Port1 Router 1 In] talk dump
```

Trigger #3 Delay Ramp up signal on

```
trigger ae1 d[Port1 Router 1 In] bus 24 on

      cmd ae1 bridgelamp 1 on ~ turns on start lamp to show delay is active
      cmd ae1 bridgelamp 2 flash continuous ~ flash dump lamp to show delay is ramping
```

Trigger #4 Delay Max signal on

```
trigger ae1 d[Port1 Router 1 In] bus 26 on
      cmd ae1 bridgelamp 2 on ~ sets dump lamp as steady on to show it's at max
```

Trigger #5 Delay Dump signal on

```
trigger ae1 d[Port1 Router 1 In] bus 26 off
      cmd ae1 bridgelamp 2 flash continuous ~ sets dump lamp to flash to show it's building
```

Trigger #6 Delay Ramp down signal

```
trigger ae1 d[Port1 Router 1 In] bus 25 on
      cmd ae1 bridgelamp 1 flash continuous ~ flash to show delay is ramping down
```

Trigger #7 Delay Off signal

```
trigger ae1 d[Port1 Router 1 In] bus 25 off
      cmd ae1 bridgelamp 1 off ~ turn off start/stop lamp
      cmd ae1 bridgelamp 2 off ~ turn off dump lamp
```